
These notes are for use with a standard Sculptor Version 6 training course.  The training course is intended for programmers who are competent in one or more languages already and who understand the basics of data file structures.

Sculptor 4GL originated in the 1980’s as a general-purpose application development system.  It was used extensively in the Unix environment and was particularly valued by developers at that time because it was possible to develop a program on one platform and run on any other platform supported by Sculptor – a key advantage at a time when there were 20 or 30 Unix variants in common use. 

In the late 1990’s Sculptor moved into the GUI and Client/Server environment. A typical Sculptor application today has a Microsoft Windows front end with a Linux or Microsoft Windows Server.


[bookmark: _Toc30515401][bookmark: _Toc82954774]A note on the development environment 

Sculptor developers work from a DOS shell on Windows or a unix /linux shell. This course assumes that the programmer is working from a DOS shell and has a text editor available. 
 
This course will gradually introduce each of the main elements of Sculptor programming and incorporate them into progressively more advanced programs. In many cases a given exercise is an upgrade from the previous exercise. 

It is useful to have the Sculptor online manual to hand. The exercises and examples in the text will often discuss only the basic form of a command or structure. If something more is required the online manual is the first place to look for information.

The online manual usually lives in C:\Sculptor\help\sc6ref.chm
[bookmark: _Toc82954775]Note on exercises 

The solutions directory contains solutions to all exercises by chapter number - in the unlikely event that you cannot do a particular exercise . The programs are named to easily identify where they came from. Eg solutions/chap02/ex1.r is the first exercise of chapter 2. Your working directory contains empty sub-directories for each chapter and it is recommended to follow this convention in naming your own solutions. 

Call up a DOS shell and change into the directory.  

			C:\SCV6TRAIN              /* or where you put your base */

In this directory you will see the following subdirectories :- 

Work – empty work directory  to use with the chapters
Notes – These notes etc
Exercises – Example solutions to all exercises 
Data – Test data for use with the exercises 
Samples – sample programs 

[bookmark: _Toc82954777]Exercise 1 – ex1.r 

Change directory into the working/chap01  directory and use your editor to create a file called ex1.r containing the following two lines of text which shows our first two commands :- prompt and exit 

			prompt “Hello world”
			exit 

Compile the program with scc 

			scc ex1.r 
		
Run the program with srepwc 
			srepw ex1

Note :- Sculptor source code is case sensitive . 
	prompt “hello” 
Is correct. But the following would get a syntax error from the compiler :-

	Prompt  “hello” 

Note : Sculptor uses double quote to specify text strings . 

	prompt ‘hello world’ 
would get unpredictable results. 

Sculptor is a semi-compiled or p-code language. When a program , ex1.r say, is compiled by scc the output from scc is ex1.q . The ‘q’ file is then run  by the Sculptor run time time system , not directly by the host operating system. The program srepw is known as an interpreter because it interpret .q files. 

The prompt command in its simplest form is one of several commands that put out a message and wait for a mouse click or RETURN key . Two others of note are info and error . The only difference in this case is the appearance of the message window. 

Historic note :- In versions of Sculptor prior to version 3 a program such as the above would have been given the file name ex1.f ( not .r ) and run with sage ex1. Reports and non-interactive programs were all given a “.r” extension and run with a program called sagerep . From Version 3 onwards it is normal to give all programs a “.r”  extension and run with srepw. But for compatibility , it is still possible to call programs with a .f extension and to run them with sagew if the you choose.  It is better however to use srepw throughout because it does all that a .f with sagew will do plus the report capabilities of a “.r” program, plus a whole lot more. 

Eg we could have had our source code in ex1.f . Then :-

            scc ex1.f     /* compile */
	  sagew ex1     /* run */ 

All common languages allow programs to have comments in them. In Sculptor, a line commencing with a dot is a comment and a block of test surrounded by /* and */ is a comment. 
Eg Our program above could have looked like this :

/* 
            simple program to say hello 
*/

	prompt “hello”
	exit

[bookmark: _Toc82954779]Chapter 2.  Essential constructs of  Sculptor 4GL language which are common to most languages

This section deals with basic language constructs which are common to just about all high-level languages. Experienced C/C++/VB programmers should be able to read this section quite quickly. Everyone should do the exercises to make sure that the basics are really understood.
Sculptor version 2 programmers with recent experience  may skip or skim  this section after noting that labels now have a colon on the end and the ‘then’ in  if  then else constructs is optional and after reading the section on functions. 

Inexperienced programmers or programmers who have not done any programming for a while should read this section carefully in conjunction with the sample code fragments in the directory SimpleProgs.  It is useful to look at the source code in this directory for each example and to try to work out what the program does before running it. 
For inexperienced programmers or beginners,  this  section could be quite difficult and time should be taken over reading and experimenting with the examples.
All the programs in this section run in a sort of console mode , ie without use of windows and with minimal screen capabilities.  This does not look particularly nice but it saves us getting into windows programming until we are ready so as to concentrate on basic language constructs only. 

[bookmark: _Toc82954780]2.1 Variables or temps and the assignment 

The simplest variable is declaration is  :-

		!temp varname,[ "Heading"], data-type 

Examples 

!temp StockCount ,   "Stock Count" , i4                       
!temp PackSize     , "Pack size" , i2
!temp editflag       ,"Has file been edited", u1 
!temp CustName , "Customer Name" , a40 
!temp x                 ,, r8  

The basic most Sculptor data types are :-

Integer – u1, i2, i4
Real    - r4, r8
Alpha  a (size) 

To assign a value to a variable simply use “=” as with nearly all languages . 
Eg

	StockCount = 1000
CustomerName = “Fred Scuttle” 

	x = ( x * 4 ) / ( x+1 ) 

The sizes you use indicate the number of bytes the variable will use, not the number of places displayed on screen except in the case of alphas. 

The names which you chose for variables do have rules but if you adopt the same rules as you apply to file names , except spaces not allowed, you should have a valid name.


Arrays are declared using square brackets

!temp CustAddress “ Address Lines “, a40[5]

The elements of arrays are accessed in the natural way , eg 

CustAddress[1] = “123 Somewhere St”
CustAddress[2] = “London”
CustAddress[3] = “NW1”


See the appendix on data types for a more exhaustive  list.  

Variable are given initial values at the point of declarations appropritae to their type. Hence numbers are set to zer and strings are set to spaces. 

We have already seen the prompt command which issues text to the user and then waits for a return key to be pressed. This command has many variations and elaborations as do most commands. 
We will use two simple variants  of the get and put commands to help us review the most basic commands of the language. 

	put “Hello” 
	put “This is line 1” , “This is line 2” 
	put “Enter a number “ ; 

Will simply write the message to the screen. The first two variants  and scroll one line down after each separate text string but the third does not because of the semi colon .  

	get myvariable 

will wait for keyboard input followed by the return key and then store  the input in myvariable .

Notice in the prompt command in getvar.r that we can have more than one textstring in the prompt command if we use comma separators. This applies generally to many commands , in particular to get and put .

For an example take a look at the SimpleProgs\getvar.r . As with all the following examples examine the source, compile it and run it 

scc getvar.r
srepw getvar.r 

The form of the if /else construct in Sculptor is much the same as most modern computer languages

Variant 1:

		if ( condition ) {
			Statements
		} else {
			Statements
		}

The “else” part of this construct is, as one would expect, optional. 

Variant 2:

		If ( condition ) Statement  else Statement
		
Older versions of Sculptor required a “then” keyword after the condition which is still supported but does not actually affect anything. 


Example

		if  ( CustomerID < 0 ) {
			prompt “Invalid ID” 
		} else {
			prompt “ID is OK”			
		}

Curly braces are used extensively in Sculptor to indicate that a block of statement lines should be treated as a single unit. 
Example code in SimpleProgs\if_else.r . Notice that in if_else.r  we use a form of the put statement with a semi-colon on the end .

	put “enter a value  “ ;

The semi colon on the end of the put command tells the system not to do a newline after outputting the message.
 
[bookmark: _Toc82954783]2.3	Functions (Subroutines )

Look at   \SimpleProg\function1.r and function2.r closely in conjunction with what follows. 

A subroutine or function or sub-program is a self-contained , optionally parameterised , block of program code which can be called by name.  

The basic form of a subroutine in sculptor is 

!function MyName(param1, param2 … ) {
	[return [value] ]
}

When a function returns,  control in the program returns to the place from where the function was called. The return command simply exits the subroutine immediately . If the subroutine exits with a  return value clause the place where the function was used  is, in effect, replaced by this value. 
Eg
!temp x ,, r8
		x = 1 - cos( 3.142 ) 


For an example,  cos()  and sin() which are used in the example are predefined functions – you don’t need to write your own of course. But in respect of the way in which they are called and used there is no difference between a user defined or a predefined function.  

Functions which return a boolean value will be used in many of the exercises which follow. An example might be :

!function	IsOddNumber(x) {

	return ( x % 2 = 1 ) 
}

Here the expression which forms the return value is TRUE or FALSE and we could then use this function in a program such as the following fragment :

	if (  OddNumber(101) ) {
		prompt "ODD" 
	} else {
		prompt "Even"
	}
 
Functions may have their own local declarations of temp variables. The scope ( or applicability ) of variables ( or temps ) declared inside a function is strictly local to the function . If a program has a variable called X declared outside any functions, and a function also has a local variable of the same name , the local X is used inside the function ( or you could say has precedence ) and has no effect on the one declared outside the function. The first X is said to have global scope , the one inside the function has local scope. 
 

You may notice that functions do not require data types for their parameters - unlike most high level languages. Sculptor will coerce data types into something sensible. In the Minimum() example above x and y may be numbers or text. If you assign an alpha field to a numeric the numeric will take on the numeric value in the alpha. 

In function2.r you will see a function called AssignDate(fred) . Notice , that if the parameter fred is changed inside the function , that value also changes in the calling routine ( the main program or another function ) . This is known as call by reference  or a reference paramter.

Sculptor supports a counted loop much like C/C++ / VB

!temp i ,, i2

		for ( i=1 ; i <= 10 ; i++ ) {
			MySubRoutine(i)
		}

This program fragment sets i to an initial value of 1, calls MyRoutine() with i as a parameter , increments i by 1, checks to see that the condition ( i <= 10 ) is true and, if so calls  MySubRountie() with i set to 2 and so on. 
The part to increment i by is given as i++ . This is the same as i = i + 1.

In general the 3 clauses inside the for loop initial line may refer to any variables ,although it is rare for them not to apply to the variable being used to count the loop . The point is that the 3 parts apply at 3 stages :- 

Stage 1 :  i=1    --- happens once only at the start of the loop. The assignment may be any valid assignment , not just an assignment to one.
Stage 2 : i <= 10 --- is tested every time program control is at the top of the loop including the first time. The loop stops if the condition is not met.  
Stage 3 : i++ or i = i + 1 happens each time after the body of the loop has been executed and before . 

Stage 3 does not have to be and increment of one. We could have i = i + 2 or  assign i to any expression , eg i =  i/10 


In SimpleProg\forloop.r 

  
[bookmark: _Toc82954785]2.5 do/while and while/do 

Sculptor supports both types of loop

Example

		
		while ( endflag = 0 ) {
			MyRoutine()
		} 
and  … 

		do { 
			MyRoutine()
		} while ( endflag = 0 ) 


Unlike if/else the counted and none counted loops all require separate lines between statements and curly braces to define blocks of code.

See SimpleProg/while1.r and See SimpleProg/while2.r and 

You can break out of any type of loop with the break command 

Eg 

	while ( cond ) {
		if ( ExitCondition ) break 
		[Other Statements] 
	}
	prompt “finished loop”

The break, if executed, will cause control to pass out of the loop to the prompt command. 

The continue command causes a counted or non counted loop to go to the test that is controlling the loop. 
Eg 

for (i=1; i < 10000; i = i + 1 )   {
		If ( TestCondition ) continue 

		[Other Statements] 

}

If the TestCondition is true the [Other Statements] section is skipped and the variable controlling the loop will be tested and incremented.  The same effect could have been achieved without a continue by with 

		If ( not TestCondition ) {
			[Other Statements] 
		}
 
See  the examples in SimpleProgs :-  break.r and continue.r 

[bookmark: _Toc82954787]2.7 goto

As any BASIC programmer will know a goto statement simply transfers control in a program to a given label. It is possible, but not natural, to write Sculptor programs with no goto statements at all. Generally, programs should be split up into neatly ordered subroutines and loops but where file handling and keyboard input are concerned it is often neater to use labels to transfer control around the program as execution proceeds

Example 


	if ( x = -1 ) goto ERR_LABEL 

	[Other program statements]

ER_LABEL:

	prompt “Something has gone wrong”
	exit


This type of jump should be used sparingly.  


We will see later that control can be  passed to a label without explicitly using a goto command, for example when a  command fails for some reason.

See the example in SimpleProgs
 
[bookmark: _Toc82954788]2.8 switch statement

This construct, familiar in some form in most languages, is the equivalent of a set of if/else statements. If there are a whole set of values to test it is easier with a switch. 

Eg 

		
	Assume an alpha field, tname, which currently holds the value "Sandra" 
	
	switch ( tname ) {
	case = "Sandra" :
				DoProc1()
				break

	case = "Sumitra"   :
				DoProc2()
				break

	case = "Eric    :
	case = "Bill"   :
				DoProc3()
				break


	default :

			prompt "No name set "

	}


In this case the function DoProc1() is called. Note that if we do not have the break statement the next statement following is also carried out. In the example if the name is set to "Eric" or "Bill" DoProc3() is called. If we did not have a break after DoProc1() both DoProc1() and DoProc2() would be called in the case where name="Sandra". The default case is optional and is called after all other options have failed.

See the example in SimpleProgs 

Many of the following exercises will need get , put and prompt together with other constructs used above.  In some exercises you may find it useful to use a command we will encounter properly later on :
		clearall 

This will clear the output window . 


Exercise 2.1  get, put 
Write a program to input two numbers  and output :-

The Sum      ( first + second )
The product ( first  x second ) 
The average  

Exercise 2.2  get, put, if/else , goto 

Input two numbers and output first divided by second after checking that the second is non zero . If it is zero make the user re-enter both numbers . 
When you declare the two numbers make sure they are reals , ie

!temp num1 ,, r8
!temp num2 ,, r8

 If you declare them as integers then , for example ,  3 divided by 2 would not come to 1.5 but to 1 because integers have nothing after the decimal point. 

Exercise 2.3  - for loop, assignments 

Write a program to output the first 20 terms on the Fibonacci sequence. Use one line per term or send them all to the same line. 

The sequence is made by adding the two previous terms in the sequence to get the next ( except for the first two terms which are just 0 and 1  ) 
0 1 1 2 3 5 8 13 …

Exercise 2.4  - for loop, while loop , function() 

Write a program ask the user for a number and then output a Times table requested for the number . Use a parametrised function OutputTable(n) to output the table.


Exercise 2.5 ,  for loop inside a for loop , formatting using ###


Output a compete multiplication table for the number 1 to 12 . To make this exercise nicely you will need a construct we have not yet met which formats an expression in a put statement 
	put x : “###” 

This will ensure that x is output in a field 3 spaces wide filled from the right with leading spaces. 

1	2	3	4	…
2	4	6	8 …
..

12	24	36	…


Exercixe 2.6 – for loop, arrays , rand(0) function , assignments

The rand(0) function returns a random number . Eg x = rand(0) assigns a random value to x . 
For spacing you many want to use the form 
              put x , 

which outputs x with a comma after i and no line feed . 


i. declare an array   
               !temp nums ,, i4[10]  

ii. Fill the array with 10 random numbers

iii. output the unsorted array . For this write a function OutputList(msg) which takes a string parameter msg  . Called it as OutputList("unsorted list" ) 

iv. Sort the array in ascending order 

v. Output the now sorted list again. 

Exercise   Special Challenge – If you have time to spare (one day) and you have done all the other exercises , try outputting this sequence .  

   1 
   1   1 
   2   1 
   1   2   1   1 
   1   1   1   2   2   1 
   3   1   2   2   1   1 
   1   3   1   1   2   2   2   1 
   1   1   1   3   2   1   3   2   1   1 
   3   1   1   3   1   2   1   1   1   3   1   2   2   1
    ....

The object of this section is to create the simplest possible windows program that will allow us to perform screen I/O and respond to a button press.
 
The basic building block of a windows program is the window itself. A window contains controls ( buttons, textboxes, listboxes , scroll bars etc ) . Windows controls respond to activity, known as events,  such as scrolling or mouse clicking by calling functions that are attached to them.   

[bookmark: _Toc82954791]Using the Sculptor screen designer - spd 

spd is the Sculptor screen designer. It creates and edits windows controls by editing the source code of your program. Therefore, you should be using a text editor or spd on your program code at any one time but not both at the same time. You can do some limited code editing in spd but we will be cutting and pasting from one program to another so it is best to use your favourite editor at least for that purpose.

If you are happy to use spd throughout you can create functions and edit them in spd and never use an external editor. From the file menu you can compile and run a program. 

To edit a function, select it from the select menu and then open the edit window in the view menu if it is not already opened. 

Change directory into working\chap03 for the following exercises.

You can create a simple program without typing in any code as follows

Type
		spd 

The from the file menu chose new program . This will create window which you can move around and resize in an intuitive way. If you save the program to , say, "basic.r" and then exit spd you will find basic .r in your default directory . 

This basic program will be the starting point of many of our early exercises. If you have your own customised  variation on basic.r  then so much the better as  long as your version has the basic features described below.    

As a point of interest,  spd uses c:\sculptor\default\default.r to load basic.r . There is also a copy in working\starter\basic.r 

listing of basic.r 
!include <sculptor.h> 
!chargrid OFF
+window wintask at 10,10 {
	max_width  = 500
	max_height = 500
	title = "Skeleton Program 1.0. Press ESC to exit"
}
	MainProg() 
	ExitProgram()

!function MainProg() {
	display wintask
	dialog wintask cancel = MAIN_RET
MAIN_RET:
	return 
}
!function ExitProgram() {
	exit
}



The 1st line  includes <sculptor.h>  , that is to say the text from "sculptor.h" is , in effect, copied into your source code at this point. The angled brackets indicate that the file is a "standard include" file located in the Sculptor include directory. On Windows this would typically be c:\sculptor\include on linux /home/SCULPTOR/include.  The standard Sculptor header contains lots of useful things that we are not making use of here but which will be essential as we progress so it is a good idea to start including it in all of your programs from now on. 
Lines such as !include <sculptor.h> are directives to the compiler as to how your program should be compiled. The included file may contain program statements or indeed anything allowable in the program file at this point . By convention however ".h" files contain definitions and declarations not code for execution. 

The line 
 !chargid 

should be noted.  All your programs intended to run under M/S windows or WINE on linux should have this compiler directive. It tells the system to use pixel positioning as opposed to character positioning when referring to the position of a control . If chargid is OFF then 

+window wintask at 10,10 

means the top left corner of wintask is at pixcel 10,10 . However if chargrid is ON the 10,10 would mean 10 characters across and 10 character positions down. On a typical linux dumb terminal a screen is 80 x 24 characters .  

The lines commencing +window winstask  completely define a  window .  All the properties of this window are held between the curly braces  ( "{  ...  }" ) , except the x,y co-ordinates of the top left corner of the window which are given in the at 10,10  sub-clause . 

The window known as wintask is a special one but we could have called our window given above by any name.  However,  if you do not declare wintask yourself it will be implicitly declared for you and opened at program startup.  Many Sculptor constructs which have an optional window parameter will default to wintask unless you explicitly mention some other window. 
The +window sub-clause indicates that the window opens at program startup . If you would like it to remain closed until you chose to open it, declare it with , for example 
 -window MyWindow at 10,10 { 
}
 
The next  two lines form the  substance of the program code . MainProg() is a user defined function, call it anything you like. It is the most important function in the program merely because it is the first executable command that srewp comes across when executing your program – all that precedes it is setting up the environment and resources for your program to run in.  

	MainProg() 
	ExitProgram()


 Look now at the lines in basic.r in the function MainProg() :-
		display wintask
		dialog wintask 

The display is needed if wintask contains any data and we wish it  to be displayed prior to the next command. Our wintask so far has no data but it soon will do.  

The dialog command gives control of the program to the controls attached to wintask ( if there are any)  . We could have declared some other window and done the dialog on it, there is no special connection between the dialog command and wintask unless the programmer chooses to make one. 


Our window , at present, has no windows controls – it is just an empty window with nothing more than a title. In particular it has no exit button.We must set that up if the usrs is ever to get out of the program ,  hence the optional sub-clause cancel = MAIN_RET which ensures that if the user presses the ESC key program control switches to the user defined label "MAIN_RET" .  
After the dialog has exited, the function returns , the next statement is to call the user defined function ExitProgram(). 

We already know about the exit command which just terminates the program and is the only command in this function. A more realistic program might have a number of things to do or to tidy up before a program exits.





 The general outline of a Sculptor program is , broadly, the sections below :-

Section 1 
	definitions    - to be discussed later
	!include files
	directives like !chargrid 
	declarations of variables and data files 

Section 2
	Windows declarations

Section 3 
	Initial program code to get things started 
Section 4 
	Your user defined functions 

For the remainder of the course it is recommended that you use basic.r as the basis for all your programs , or as things develop just use a previous program to continue with the next. 

The command line call to spd on your program, "myprog.r" is what you would expect :
		spd myprog.r 

spd will show wintask , as it currently would appear to a user and wintask as it appears in your program . Close the window marked 'Edit window' , which shows your program code -  we do not need it for our present purposes.
From here you will see various menus and options. The screen editor is intended to be mostly intuitive and so this we will not attempt to describe it in detail.  In the section we will concentrate on :-
	creating a text box
	creating a button 
	Attaching an event to a button 

Points to note about spd :

You can move controls around with drag and drop and resize – as one would expect. 
If you click on a control a properties box appears on the right. Some properties have a drop-down list to choose from others just a space for text. Some controls, eg font, allow you to type in a question mark which then offers some options. 

spd does not allow you to edit your program code except in a very limited way.  It is best to save and exit to do that in your favourite editor.

Try the following out from the create menu at the top left of the spd screen :-

Create a text box . Take a look at the properties of the textbox. You may need to click on it first if the properties panel is showing properties for wintask or no properties at all. The textbox has a default name, Textbox1,  which you will use later in commands such as display and clear but it is not associated with any data. You may call the textbox anything (sensible) that you chose. You will need to fill in the field which supplies a data value to the text box. We don't have any data in our program yet so we will use a pre-defined value , sys.Date , to fill in the field property. You will see your text box acquires a default label  "Date" which is a default label for sys.Date . 

Create a push button . You will be offered options for Top level and In group. Chose top level for now ( more on groups later ) . Give your button a label . 
Notice the event and event_enable properties. These are available for most controls but are particularly of interest in relation to buttons because there is not much point having a button that never has the chance to do something. The event property should be the name of a function which is called whenever anything (absolutely anything) happens to the button. For most events occurring in the life of a button it is not required to call the event function. For example,   at program startup the button is created and the event of creation could call the event function. We can control when the event function is called by listing the events we are interested in in the event_enable property of the control . 

If you put a question mark in the event enable property you will be offered a number of possible events for the control. A typical choice for a button would be BUTTON_CLICKED.  Sometimes you may want more than one of the options offered , use that standard ctl+click to select more than one option.

We have only the ExitProg() funciton in basic.r at the moment so attach this function to the button and give the button a label "EXIT" , say. 

It is not just buttons which can have event functions attached to them .  A window for example can have an event function. One of the important events in the life of a window is the CLOSE event ( press ? in the event_enable property to see it ) .  If we wanted to hook up our ExitProg() function to wintask so that we could close the program by pressing a standard close window button on wintask we would need to do the following :-
1. Edit the style property of wintask to have a close button ( ie the small cross in the top right of most windows ) 
2. Edit the event and event_enable properties as we did above for a button and attach ExitProg() 

 
[bookmark: _Toc82954793]Summary of important points 

1. The "+window" part of the window declaration indicates that the window is to be created and opened immediately on program start-up.
"-window" indicates the window is closed on start-up. 
2. In general it is best to use spd strictly for creating and editing windows controls. Declarations of variables and files etc is best done in your editor. There is no programmer's work bench or development environment. Therefore, once again, use your editor or the screen painter but not both at the same time. 

3. Notice that many of the properties associated with a control are set with a list box. For those that are not it is often worth type a question mark if you are not sure what entries are available. 

4. Use a clean copy of basic.r as a starter for your programs. You can get a clean copy from working\basic.r or use the new prog option in spd. 
. 

[bookmark: _Toc82954794]Exercises 3 

Exercise 3.1 dialog, window style and events, user defined function

Write a simple program which  displays the system date in a textbox and allows the user to exit the program by clicking on the close button of the window. 
Use the style property of wintask to add in CLOSE , MINIMISE and MAXIMISE controls to the top right of the window. 
Attach the ExitProgram() function to wintask as its event. Make sure to chose a suitable event_enable property – EV_CLOSE.
Exercise 3.2 dialog, push button , put 
Update exercise 2.4, to output a times table, by creating a textbox for the input parameter for required table. The output should still go to the screen using a put as before. 
 
You wont need or want to pass a parameter to your output table function. The data attached to Textbox1 will be used throughout. 




[bookmark: _Toc82954795]Chapter 4 – The Sculptor filing system

Sculptor 2 programmers should be able to skim this chapter quite quickly and go to the exercises to check that they have not forgotten anything. Those with no experience of Sculptor will need to proceed quite slowly and carefully. 

Sculptor has a filing system – not a database. Sculptor data can be accessed by ODBC but this section deals exclusively with using Sculptor language to get to the data.

[bookmark: _Toc82954796]Basic data file structures : Files, records , fields and the file buffer 
                                           
What follows is a general round up of basic file concepts and how they relate to Sculptor in the most abstract way. Specific commands are dealt with later, suffice it to say for now that we have ways to write to a file and move around it.

A file is a set of records. A record is a set of fields. In Sculptor we deal with one record at a time.  The file buffer is a copy of the record in the file that we are currently looking at. If you change it you must still remember to write or re-insert it into the file.

Your working directory has a sub-directory called working\data which contains a number of data files including the one we are about to look at – stockitem 
 



Each record in stockitem has the fields shown. We will look at how you find out characteristics of the fields in a data file in the next section .  For now just assume we have the fields shown and they have sensible data types and sizes. 

   Code        - stock code    
   Desc         - description 
   Weight     - weight 
   Price       - price
   SDate       - Start date of product line 
   InStock    - number in stock 


Here are a few records from somewhere in the middle of stockitem  :-

                                    Current Record Code : SPANN-001
desc :Short type
weight : 3.0
price :  5.0
sdate : 1/1/2021
Instock  : 1000




  code Code : SPANN-003(T)
desc : Curved spanner
weight : 25.0
price :  17.50
sdate : 1/3/2021
Instock  : 98




Code : SPANN-002(X)
desc : Long type
weight 20.0
price 15.99
sdate : 1/2/2021
Instock  : 250





   desc       
   weight     
   price      
   sdate      
  

    	


	Working file bufferCode : SPANN-002(X)
desc : Long type
weight 20.0
price 15.99
sdate : 1/2/2021
Instock  : 250












The big arrow is the file pointer and it points at the current record in the physical file on your hard disk. A program accessing this file has a private copy of the current record known as the file buffer. The file buffer is changed every time the file pointer moves. If we change the buffer it does not change what is in the file until we tell the program to put the buffer and data file in sync. If we change the file buffer but do not update the file with the changes and then move the file pointer - the buffer changes would be lost.
 
If we wanted to put out a prompt out with the value of the number in stock and the price for item code SPANN-002(X) we would simply say 	
	prompt stockitem.Code , stockitem.Price

and this would display the two values of the file buffer.  

 If we wanted to see fields belonging to the third record shown here, SPANN-003(T), we would need to move the file pointer forward first, using an appropriate Sculptor command, and that would update the file buffer to be a copy of the new current record. 

If we wanted to change the value on file of the price of SPANN-002 we would do in a two stages 
1. Set stockitem.Price = NEWVALUE ( and other changes )
2. Write the file buffer , which is only a copy of what in is the file, back to the file. 
So , after step 1 the buffer and the record being pointed to in the file have a different value for price. After step 2 they are back in sync.


[bookmark: _Toc82954797]A first look at indices 

You will notice that in the stockitem snippet give above, the records were in the order determined alphabetically by the field called code .  For stockitem,  this is the only key field and it means that when we traverse the file the records always come in ascending key order sequence , or in code order sequence. If we had made , say, instock as the sole key the order of the records in the diagram above would have been reversed. 

A data file index can have more than one key and a datafile can have more than one index which we encounter in the section on secondary indices. 





[bookmark: _Toc82954798]The three physical files that make up a conceptual data file . 

A Sculptor data file structure , called ‘stockitem’ say,  comprises 3 separate physical  files  
	
		stockitem – The data itself
		stockitem.k – an index into stockitem
		stockitem.d – a data dictionary

Generally, programmers are unaware of how Sculptor organizes its data – they simply rely on the filing system to work in the way indicated by the diagram below . Conceptually these files should be thought of as one.  

A data dictionary is a list or catalogue of the fields which make up a the records in a data  file with their types and sizes and other characteristics such as whether they are key fields. 

[bookmark: _Toc82954799]Four important utilities : pdes, kfcheck, newkf, ddeditor 

ddeditor data-name  -  Sculptor data dictionaries are edited with this  program                                                             which is itself written in Sculptor.  Its most important function, amongst other things, is to change the record structure of a data file. 
pdes  data-name      - list the fields of a data dictionary 
kfcheck data-name   - Check the integrity of a file and its index(s)

newkf   data-name   - Initialise a Sculptor data file. Any data in the file is lost, so be careful !!!

[bookmark: _Toc82954800]Programming using the Sculptor file system 

A file which is opened automatically on program startup is  be declared in one of the 3 ways shown.

!ofile stockitem “../data/stockitem” update
!ofile stockitem “../data/stockitem” read
!ofile stockitem “../data/stockitem” create 
 
Be careful with the "create" form - this initialises the data file and deletes any data in it in the process !!!!

We will use only  one of these forms for now - however the use of the openfile command ( below ) will become more natural as we progress

[bookmark: _Toc82954801]A note on slashes and backslashes 

Sculptor is usually happy to mix forward slashes and back slashes in many contexts. Forward slashes are preferred in file names  because it makes your code portable across Linux ( which always uses forward slash ) and windows which always uses back slash. In some contexts you will need to put two back slashes when you only want one because back slash is an escape character – meaning take the next character literally -  even if it has special meaning. Examples  

testname = "Fred Bloggs" 
testname = "Fred \"The Hat\" Bloggs" 
testname = "ted \\teddy bloggs" 

In each case we use a back slash initially to  escape the next character. In the second assignment we need it because the double quote would be taken for the end of the string. In the 3rd case \t would be taken as a control character for tab if the preceding backslash did not escape its normal meaning.  

The 8 fundamental Sculptor file access commands are :-

next,prev,read,find, match, delete, insert, write


[bookmark: _Toc82954802]The next and prev commands

Update the file position to the next record in key order .

Eg
	next stockitem 

After a successful next, or any file pointer movement,  the file buffer of stockitem contains a copy of the required record. 

Eg

	next stockitem 
	prompt stockitem.Code

Notice that we can qualify the field  with the name of the data file but this is not mandatory.  Ie we could just say : 
prompt code 
However it is best to qualify in case we have more than one thing called code .

[bookmark: _Toc82954803]The system variable sys.Traps 

sys.Traps is an integer which is set after a file operation such as next . If the value of sys.Traps is zero - all is well . Otherwise there was an error. In the case of next, for example, the error could be that we have reached the end of the file.  The zero value also has the symbolic name OKAY and it is normal to use this form. 

Any file command can fail so we therefore need to trap errors. If the command itself does not do this the system will do it for you – though not in the most elegant way. Therefore  it is best to tell your file access command to ignore errors and let you check the system variable sys.Traps yourself. 

Eg

	next stockitem traps = IGNORE
	if ( sys.Traps <> OKAY) {
		error “End of file reached” 
}
 

The prev command, meaning get the previous record in key order, is identical to next except for the direction of key traversal.

[bookmark: _Toc82954804]The read command 

The read command  reads a record from a file using an assigned key. Like next and prev, read moves the file pointer hence subsequent next /prev operations will be relative to the position last read from. 

Eg

	stockitem.Code = "SPANNER"
	read stockitem traps = IGNORE
	if ( sys.Traps <> OKAY ) {
		prompt "Cannot find SPANNER"
	}

Notice that read requires the whole key. Thus if "SPANNER" was in the file but we searched using read on "SPAN" the command would fail. We will look at key structures in more detail later. For now, note that stockitem has only one key, stockitem.Code. If it had 2 or more keys , all keys would have to be supplied the make the read command succeed.

In the form we gave above , the read command needed two steps to read the required record. A shorter and usually better form would be :-

	read stockitems key = "SPANNER" traps = IGNORE 

Had there been more than one key just separate the values by commas. 

[bookmark: _Toc82954805]The find and match commands  

The find command is similar to the read command but is less exact about what it matches to the supplied key. 

Eg 
Assume stockitem has just three codes begining "SP"  :-

		SPANN-01
		SPANN-02(X)
		SPANN-03(T)

and we then give the command:-

		find stockitem key = "SP"

The file buffer would be set to the record with "SPANN-001" as a key because that is the first "SP" in the file.  The match criterion being applied here is the we are searching for a key starting with "SP" and continuing with anything after the first two letters. 

To get to the other records in the file which also match the search criteria we use the match command which just continues the find where it left off.  Thus :-

		find stockitem key = "SP"
prompt stockitem.Code
		match stockitem 
		prompt stockitem.Code
match stockitem 
		prompt stockitem.Code

This would display each of the values SPANN-001, SPANN-002(X) and SPANN-003(T) in turn. 
 



A typical find/match loop might look like this where we process all the S's in the stockitem file with a user defined function that does something ( we don't care what – within reason ) .  

	find stockitem key = "S" traps = IGNORE 
	while ( sys.Traps = OKAY) {
		DoSomethingWithTheRecord() 
		match stockitem traps = IGNORE
	}


[bookmark: _Toc82954806]The insert and clear commands 

The clear command occurs in a number of contexts in the Sculptor programming language. Our interest at present is clear applied to a file buffer.

		clear stockitem

This command has no effect on the data file stockitem - it only clears the fields of the file buffer. The cleared values of fields are the natural ones you might expect, alpha fields set to spaces, numerics to zero, dates to zero . Before inserting a new record into a file it is a good idea to clear the file buffer in this way even if you think all the fields are about to be set to an initial value by later statements in your program. It would be an easy mistake to have junk values in a file buffer left over from some previous operation affecting the values written when you insert a new record. 

To insert a new record into a file, clear the buffer , assign the fields, insert into the file.

Eg

		clear stockitem
		stockitem.Code     = "SPANN-004(Z)"
		stockitem.Desc     = "Pink handled spanner"
		stockitem.Weight   = 123.45
		stockitem.Price    = 2499
		stockitem.SDate    = "1/1/2006"
		stockitem.InStock  = 0 

		insert stockitem traps = IGNORE
		if ( sys.Traps <>  OKAY) {
			error "This is already on file"
		}


Note that the money field, price, is set in pennies not pounds. On interactive data entry the price above would be entered in the natural way as 24.99 but inside programs we work in pennies, eg 2499 
An insert will fail if the key value(s) of the record already exist in the file or if the file was opened in read-only mode. Sculptor does not allow multiple identical keys. 

 
The delete command

To delete a single record you must first of all have the intended record in the file buffer using a next/prev/find/match/read the record. Make sure the file is open in update mode and then issue the delete command. 

Eg  read and then delete the record previously inserted 

		stockitem.Code = "SPANN-004(Z)"
		read stockitem traps = IGNORE

		if ( sys.Traps = OKAY ) {
			delete stockitem traps = IGNORE 
		}

		if ( sys.Traps <> OKAY ) {
			prompt "could not delete record"
		}

We have said nothing so far about multiuser considerations, we will be dealing with that later. For now, it is worth noting that if the read command succeeds the delete command should also succeed. If the read command fails it could fail either because the requested record does not exist or because the record is locked by another user. 



[bookmark: _Toc82954807]The write command 

To change the value of an existing record, locate the record with one of the next/prev/find/match/read commands, assign a new value to one or more fields then issue the write command.

Eg

		 
		stockitem.Code = "SPANN-002(X)"
		read stockitem traps = IGNORE

		if ( sys.Traps = OKAY ) {
			stockitems.instock = stockitems.instock -1 
			write stockitems traps = IGNORE
		}			

		if ( sys.Traps <> OKAY ) {
			prompt "could not update the  record"
		}
 
A write command will fail if a current a record has not been set, if the file is not open in update mode or if a key has been changed to an existing value.

Eg

	/* assume SPANER and WRENCH both exist on the file */
	stockitem.Code = "SPANNER" 
	read stockitem
	stockitem.Code = "WRENCH" 
	write stockitem traps = IGNORE    /* this write fails because the key exists */

[bookmark: _Toc82954808]wind/rewind commands 

Two other commands which update a file position are often useful

Eg

	rewind stockitem     /* rewind a file to just before its start position */
	wind   stockitem     /* wind a file to just after the end position   */ 

It is nearly always necessary to do a next or prev after one of these commands - the exception to that rule is dealt with later. 

 
[bookmark: _Toc82954809]Chapter 4 Exercises 

Exercise 0 .
Before doing the exercises change directory into working\data and try out pdes , kfcheck and ddeditor on the data files . The First two utilities are read only and simply provide information about the data file. If by some misadventure you damage a file there is a copy in working\data.bak . You should not need the backup just yet – unless you run newkf by mistake !   

Exercise 1  next,prev, spd, pdes , display 

i. Copy basic.r to chap04/ex1.r 

ii. Edit ex1.r to declare ../data/stockitem and open it read only mode . Create empty functions called evNext() and evPrev()  
Exit the editor.

iii. Use spd to create top level textboxes for each of the fields belonging to stockitem.d . You can get the fields names from pdes but you can also get them in spd by clicking on Object Tree and expanding the stockitem definition.  However the best way to fill in the field property will be to type ? in the field property – it will offer you all the temp variable and file buffer fields it know about.

Create buttons labelled Next Item  and Prev Item  and hook then up to the two functions you created.  Remember to filter the allowed events with the event_enable property. 

Put  close, minimise and maximise buttons on wintask using the style property of wintask 

Exit spd 

iv. Fill in the code to respond to the push buttons.  After you have done a next or prev the values in the file buffer will be updated to the data in the next/prev record but the textboxes on screen will not be updated. Use display to redraw wintask. 
Remember next and prev can fail. Handle this. 


Exercise 2 – wind/rewind

Copy ex1.r to ex2.r and add in buttons for wind and rewind. You should be able to create the buttons quickly by right clicking on an existing button and making a copy. Then right click on a place on the window to paste. Then edit the properties of the button. 

When you write the event functions remember to clear wintask – otherwise you may have data on the screen which is not the data in the file buffer . 

Excerise 3  next/write/rewind / goto 

Create an option to Put all the prices up by 10% . 

i. Create a new function evPriceInc() using your editor
ii. Use spd to create a button attached to the function 
iii. Fill in the details of the function 

You can just copy ex2.r to ex3.r and add a button for Price Increase . 

Check that your program did the updates as expected when you take a look at the  the prices of the first few stockitems. 

When you use the write function make sure you trap any errors . What could possibly go wrong with a write function – assuming the file pointer points to a record we want to write to ? ( Hint :- check the file declaration ) 




Exercise 4 – find, match
i. Copy ex3.r to ex4.r  to use for this exercise . Add a Function evCountPrefix() which will work how many part codes in stockitem have a given initial few characters . The required prefix will be the one currently shown in wintask. It can be edited to the required value before you press the button. 

For example there are 48 stockitem codes starting with the value "REG" 

If you find any records display the last one found otherwise clear wintask. 

ii.   If you set stockitem.Code to blank , how many values would you expect to there to be ? 

iii. If you searched on "reg" you would get no matches but "REG" gets 48 matches.  Hence it is necessary to press the caps lock when you type in the search key. However, all the keys in the data file are upper case so we may as well set the data dictionary to indicated that this filed is upper case , then when you type into this field on-screen you wont need to press the shift or caps key. 
Use ddeditor to do this . 

Exercise 5 – insert 
Make a copy of ex4.r to ex5.r and add  button "Insert" which will insert the record currently shown on the screen in to the file . Make sure you trap any errors . What happens if you press the insert key twice without changing the data on-screen ?  Why ? 

What does a successful insert operation do to the file pointer ? 

Exercise 6  ( delete )

Add a delete record option which deletes the record you are currently looking at – if there is one. Make sure you trap errors .  If the delete succeeds move on to the next record and display it. If the deleted record was the last record in the file, clear wintask.


By running your program check out what happens if you try to delete a record before the start of the file or after the end or after a read has failed. 


Exercise 7 - read, next 

Add a read option which looks up and displays the exact stockitem code shown in wintask – if it exists. Otherwise use next to go on to the next record and display that if it exists.




[bookmark: _Toc82954810]Chapter 5 Windows programming (1)

So far, a dialog has meant a dialog on wintask and exiting a dialog has amounted to the same thing as exiting a program.  As you might expect you can have as many windows as you want and as many dialogs as you want. 

 In what follows we will learn how to set up a menu and from spd and  call  dialogs on other windows . 

[bookmark: _Toc82954811]Menus

Nearly all the work of setting up menus is done through spd. From the create menu choose "Menu" and give your menu a name. The menu will belong to the current window and will not be accessible except when this window is active. 
 Next,  create one or more menu items. Once you have created a set of menu stubs click on a menu item to edit its properties. Shift click on any menu item to get to the menu editor which can be used to add/delete menu items and move them around. A complete demonstration of screen screen designer aspects of working with menus will given in class – it is much quicker to show than to write down. For further information look in the online help that comes with spd. 

For a menu to do any useful work the menu items it comprises must be hooked up to functions – just like buttons. If you select a menu item you will see a property called Select function that does the job of making the link to a function. There is no need for an event enable since a menu function is only ever called by clicking on it, ie it only has one event. 

 
In your program you can create any number of menus. By default none of them are active and you must explicitly open a menu for it to work. This will be done with the "open" command called just before a dialog. 


Eg

	open MyMainMenu
	dialog wintask 

If main your program just opened a menu and then did nothing else it would not run the menu – it would just exit. A dialog needs to be running for an open menu to be accessible. 

In the working directory working\chap05 you will find a program called example.r . We will use this program to illustrate what follows and as  starter program for the end of chapter exercises. Therefore have this program open at the same time as reading the rest of the chapter , both source code and run the program on screen. 

[bookmark: _Toc82954812]example.r –  new constructs and points to note

From now on we will mostly just look at example programs to introduce new commands and features. In no particular order these are the new things we find in chap05/example.r plus sections on record structures which you will need in one of the exercises and on groups and include files. 

[bookmark: _Toc82954813]i) -window 

wintask looks similar to other wintasks that we have looked at so far.  But in the example we have declared two other  windows :-  winEditCustomer and winEditStockItem . Both do similar things to the previous exercises but in separate windows. Also, the windows are defined with a minus to ensure they are not opened by default until we want them opened. 

[bookmark: _Toc82954814]ii) Graphics , boxes, lines,  images
You will see a box drawn around the address fields in customer. This is done in spd – create | Graphic Item. By default, a graphic item is a box but if you want a line, for example, just create a box and then edit its type property to be something other than a box . You will see than you are offered horizontal lines, vertical lines, images etc. If you use the image or animation types you will also need to fill in the image property to tell the system where to find an image/animation file.

[bookmark: _Toc82954815]iii) CLOSE_DIALOG, CANCEL_DIALOG , DEFAULT_BUTTON 

If you look at the function 

 		!function CustomerWin() 

You will see that it calls a dialog in winCustomer . The dialog command 

		dialog winCustomer cancel = EC_CANCEL 
 
has a cancel clause which will cause control to pass to the label EC_CANCEL but only if the dialog is cancelled. However, if the dialog ends normally ( which we have not encountered yet ) control will pass to the next statement 
		info "Dialog ended normally"  

So far in our examples , dialogs have mostly ended because the program ended. Eg our evExitProg() was called by a close button event. The normal way to  control how a dialog ends and what happens next is to have event functions attached to the dialog window, or one of its controls, which signal to the system that the currently active dialog should end  , either wiht a normal or cancelled exit. 

winCustomer has buttons labelled "Close this dialog OK"  and "Cancel" which call event functions evOKCustomerDialog()  and evOKCancelDialog() . When an event occurs and you want to close a dialog because of it, the event function should do one of two things :
	return CLOSE_DIALOG 
or
	return CANCEL_DIALOG 

This will return control to the calling dialog statement which will then continue on to the next statement or jump to a cancel = MY_LABEL label depending on the return value. 
This is a fundamental feature of windows programming , in Sculptor as with other systems, and it should be studied closely. In example.r we have :

dialog winCustomer cancel = EC_CANCEL
		info "Normal exit" 
		goto EC_DONE 

EC_CANCEL:
		info "dialog was cancelled" 
EC_DONE:
		close winCustomer

If any event function in winCustomer  returns CLOSE_DIALOG , control in the program returns to the dialog statement and continues to the "normal exit" message.  Any event function in winCustoerm which returns CANCEL_DIALOG will cause control to pass to the "dialog was cancelled " message. 

Notice the style property of the winCustomer.Button5 which closes the dialog . It has style = DEFAULT_BUTTON . This means that if the user presses the return key the button will be , in effect, clicked . 

[bookmark: _Toc82954816]iv) parent windows , WS_NOTABSTOP , system variables

The menu item, "Show Settings", opens winSettings . There are some important differences between winSettings and the two windows winCustomers and winStockItems . 

winSettings has the property  parent=wintask .  winSettings only needs to be open for it to be part of any running dialog on wintask ( or whatever its parent happens to be )  . If wintask was closed , winSettings would close with it . winSettings stays within the boundaries of wintask as would any control defined inside wintask . ( If you look up WS_OWNED in the manual there is an exception to this which we do not cover here.)   

The fields displayed in winSettings are a small selection of Sculptor system variables . 

If you take a look at the properties of the textboxes which display these variables you will see that each has the property WS_NOTABSTOP which , as you can readily see when you run the program, prevents the cursor setting focus to any of these fields and hence prevents input into them. 

[bookmark: _Toc82954817]iv)  setfocus , enable/disable , hide/show , caption  commands. WS_INVISIBLE style. 

If you run example.q and select the edit customer option you will see a number of buttons to navigate around the customer file and to write or insert new data. 
Consider now the function evNewCustomer() . After  clearing the data in the window ready for a new record to be typed in , all current buttons are disabled to prevent the user doing something else in the middle of a new customer data entry . Two other buttons , to either save  the new record being input or to cancel it appear on the screen . For this we use the commands 
	disable  window-object(s)
	show window-object(s)

The distinction between show and enable etc should be clear .  When the user selects "Save New " or "Cancel new the program attempts to save the new data or forget it as appropriate -  check out the event functions attached to the buttons evInsertNewCustomer() and evCancelNewCustomer() .  They do the opposite to what was done in evNewCustomer() in respect of the buttons . Ie the Save and Cancel buttons are hidden and the remaining buttons are re-enabled using the commands 


enable  window-object(s)
hide window-object(s)

Notice , for example, that we refer to winCustomer.Button1 .  The qualifying part winCustomer is needed because there is more than one testbox called Textbox1 in the program. 

One last thing to notice in this section is that the two buttons Save and Cancel on entering a new record are both invisible. We could have achieved this effect by hiding them when the program started, as we do above,  but if you check out the properties of these buttons ( Button9 and Button10 )  you will see that they are actually  have a style already:-

	style = WS_INVISIBLE 
The show command countermands  this style .
  
[bookmark: _Toc82954818]v)  Validation functions , validate property , caption 
If a text box or other data entry control has a validation function property then the validation function is called whenever the value of the field attached to the control changes as a result of data entry into the control. A typical validation function will check the value of the data to see if it makes sense and, if there is an error , put out an appropriate message and use the setfocus command to force the user back in to the  control. The control winCustomer.Textbox1, which we use to enter company codes in the customer windows,  has a property 
	validate = ValCustomerCode
	
 If you look at the code for the function ValCustomerCode() you will see that customer code is check to see if it begins with a zero or an "O" . If it does then an error arises and focus returns to the text box. 

Notice the new command bell – big prizes are available for guessing what this does !
Notice the construct "bw" which stands for "Begins With"

	if ( customer.Code bw "0" ) ... 



Notice the new command caption which outputs text on the caption line at the bottom of the screen. By default the caption goes to wintask but you can put a caption in other windows by naming the destination. 
Eg
	caption "Operation OK"
	caption "Operation OK" in winCustomer 


One last thing to notice in this function is  the distinction between a text box and its data field property . We test customer.Code ( data) but we setfocus to winCustomer.Textbox1. Likewise hide/show, display all apply to a control not the data in a control etc etc 


[bookmark: _Toc82954819]vi) Parameters to events , use of object operator ?= 
[bookmark: _Toc82954820]Take a look at the function evNavigateStockItem() which is the event function for both Next and Prev  record in the stockitems window. 
[bookmark: _Toc82954821]The  event function when called, needs to know if it is being called by Button1 or Button2. You will see that the event function has parameters : evcode and evobj . All our event function up to this point have not had parameters; you don't have to declare them if you have no use for them. You could, if you wished, decalre just the first paramter and ignore the second.  They can be given any name you chose but the order is important :

[bookmark: _Toc82954822]Parameter 1 is always the event code , eg EV_BUTTON_CLICKED 
[bookmark: _Toc82954823]Parameter 2 is always the event object. In our case this will be Button1 or Button2 

There are other parameters after parameter 2 . In fact the number of parameters varies quite a lot depending on the type of control the event function is attached to.  
In evNavigateStockItem(evcode, evobj) we check evobj to see if it is Button1 or Button2 and then do a next or prev accordingly. Notice however, that we say 

		if ( evobj ?= winStockItem.Button1 )  ... 
and not
	 	if ( evobj = winStockItem.Button1 ) ... 

The equality operator for windows controls , as opposed to standard variables and fields in file buffers, is "?=" not "=" . 

One last thing to note .  In all the of buttons we have used so far we have set event_enable = EV_BUTTON_CLICKED  

You do not have to set the event_enable property . The only problem with not doing so is that the event function will be firing off every time anything happens to the button, not just when it is clicked. In fact the first thing your program does is to create all its controls which causes an  EV_CREATE event. You can check the event code that caused your function to execute by looking at evcode. 
eg if we left the event_enable property blank and wanted the event function to only respond to EV_BUTTON_CLICKED , we could say 

	if ( evcode <> EV_BUTTON_CLICKED ) return 

This is functionally equivalent to the way things are done in example.r .

evcode is an integer, not an classed as an  object like a control, so we use the normal equality operator here ,  "=" 


[bookmark: _Toc82954824]vii) !record 

The record definition is useful for grouping together a group of temp variables or making a copy of a file buffer - these 2 uses give rise to 2 forms of the declaration :-


Form 1. A group of temp fields 
Eg
	!record ReportTotals {
		TotalRecs   ,, i4
		TotalErrors ,, i4
		TotalOK      ,, i4
	}

The fields can be referred to as TotalRecs , TotalError etc or as RepTotals.TotalRecs, RepTotals.TotalErrors. etc.

We could have just defined 3 temp variables and achieved the same thing, like this :-
!temp TotalRecs   ,, i4
	!temp	TotalErrors ,, i4
	!temp	TotalOK      ,, i4


But using record structures in this way is better because in a big program with hundreds of temps it groups things together that belong together. Also, we do not need to think of ever more bizarre  names for our variables when we need two variable with a similar purpose. Eg we might also have :
!record  StockFileTotals {
TotalRecs   ,, i4
		TotalErrors ,, i4
		TotalOK      ,, i4	
	}

Out of interest, all the temps in your programs which appear not to be already part of a group  are in fact member of the record called simply tmp. So , 
!temp mystring ,, a32
could be referred to as mystring or tmp.mystring . 

Form 2 – a copy of a file buffer or other record 

For our immediate use we will use the form which makes a copy of a file buffer. Assuming we have already declared stockitem  as a file we can declare an identical record structure in one declaration .
!record BufferCopy stocklitem 

BufferCopy is simply a record  identical in structure to the file buffer. Other than that,  it has no direct connection with the stock item file unless we program things that way and the data in the BufferCopy is not copied from the stockitem record buffer unless you do it yourself. It is usual to use a record structure to edit fields on screen  and then copy the  record to the file buffer just before a file update instead of making the fields of a record buffer the data fields of textboxes. We have been doing this this a great deal but now we know better and it is time to stop. 
 
Eg

!ofile stockitem "../data/stockitem" update

!record CopyStock  stockitem 


	clear CopyStock
	CopyStock..Code = "XYZ"
	CopyStock..Desc = "Very useful tool"
	CopyStock.Price =  1000000

	ETC 

Then just before updating the data file, for example, we can assign all the fields to the file buffer for stockitem and insert it in the file. 

	stockitem = CopyStock
	insert stockitem 

[bookmark: _Toc82954825]viii) Include files
We mentioned include files quite early on in basic.r . It can very very useful to split the text of your program into separate include files. Sculptor does not provide for separately compiled modules and most developers have sets of library functions which they maintain in isolation from each other and can be included in any number of programs. The contents of an include file can be anything – declarations , windows definitions, functions etc etc 

We could update  basic.r ( from our first look at spd )  to read like this :-
 
 !include <sculptor.h> 
!include "basic.h" 

	MainProg() 
	ExitProgram()
!include "basic.i" 

If we create "basic.h" so that it contains all the window definitions of basic.r and "basic.i" so that it contains all the functions of basic.r then we have made our source code look a lot neater and more manageable – at least in the case where programs have started to a large number of window definitions and functions. The include directive merely tells the compiler to get all the text in the include file and drop it into the text of the program at the pint where the include is used. The complete effect on our compiled program is precisely nothing but the source code is not neatly split up . Moreover , if one or more of the include files is particularly useful and general , it can be re-used in other programs. 

It is conventional but not mandatory to give include files a ".h" extension if they contain definitions   and a ".i" or ".inc" extension if they contain program code. 


	



[bookmark: _Toc82954826]Exercises 5

Exercise 1  ddeditor, newkf 

Change dir into working/data and use ddeditor to create a file for passwords called passwd with  fields , UserName,  PassWord , AdminUser. The first field should be the only key. The AdminUser field should be of type a1.  We will use this to determine who is an administrator user by putting Y/N in the field. Chose suitable length for the other fields.  Use a "u" format for UserName , we will need this later.  
Initialise the file with newkf  - we will be using it. 

Exercise 2 – dialog, creating buttons with events 
Your working/chap05 directory already contains a file ex2.r which is example.r with a window for navigating and editing the passwords file created above. 

Improve this program by not allowing blank usernames or passwords – use a validation function for this. 
Use ex2.r it to add a few usernames and passwords with at least one of them as an Administrator . We will use the data in the next exercise.  

Exercise 3 – 
First copy ex2.r to ex3.r  . 
Create a window, winGetPasswd,  for use in an introductory dialog  which asks for a username and password .  If the user cannot provide a valid username and passwd at the third attempt then exit the program . 

In the standard reference manual, lookup EM_NOECHO . Use this to supress echo of the password when you type it in . 
 
If the user attempts to access the new Manage passwords option only allow it if the user is an administrator. 

Exercise 4 – include files 
Create ex4.r which has only the file  declarations of ex3.r and call to the main function  in it toegther with suitable include directives at the point where you need to include text.  Cut and paste from ex3.r into ex4.h and ex4.i .   The result should be the same program as ex3.r but be neater to handle. 


[bookmark: _Toc82954827]Chapter 6 Windows programming  (2) 

In this section we will extend our use of windows controls to cover radio groups, chekbox groups , textbox groups, and lisboxes. We will be working our way towards building an order entry screen. Before covering more windows controls however we will pause to look at one data construct that helps us :-



[bookmark: _Toc82954828]Groups, Textbox groups 

It is possible to group text boxes together in a single textbox group which can then be accessed as a unit. This comes in useful when using commands such as clear , enable and disable and saves having to name each textbox in the group separately. 
	eg display MyTextboxGroup

When you create a textbox with spd you are offered the option of putting the new textbox in a group if a group is not already selected. To select a group choose the group item on the "select" menu in spd . If you are about to create a set of textboxes then if you create a group first and set its type property to textbox , the subsequent textboxes you create will all be put in the group. 

If you already have a set of text boxes which are not part of a group and you wish to make them so, then you must edit the group manually as shown below. Make sure spd is not running if you do this outside spd. Suppose we have two textboxes inside a window definition. To put these 2 textboxes in a group 


		+textbox tbPrice at 325,202 {
			field = stockitem.Price
		}

		+textbox tbDesc at 246,156 {
			field = stockitem.Desc
		}

		
Add 2 lines surrounding the textbox definitions :- 

		textbox group  tbgMainText {

			+textbox tbPrice at 325,202 {
				field = stockitem.Price
			}

			+textbox tbDesc at 246,156 {
				field = stockitem.Desc
			}

		}

You can do this using the edit window in spd or in your text editor. 

 
[bookmark: _Toc82954829]Radio groups 
Refer to samples/controls/radio.r for this section.
A radio group is a set of buttons grouped together such that only one of the buttons is selected  at any one time. The member buttons need to be grouped under the same group name so that the system knows which buttons to set to unchecked when one of the other buttons is checked. 

The best way to create a radio button is to use the create | group option in spd. Set the group type to radio if it is not already set. Then add buttons to the group  using the create | radio button option giving the group name already set up if spd asks for a group.

For each button in your radio group you can optionally set the "unchecked value" and "checked" value in the properties for the button. If you then associate a field (by filling in the name of a temp or other variable)   with each radio button this field takes the appropriate checked or unchecked value as the user clicks the buttons . The values are set inside quoted strings even if the group field is an integer. 

When you need to display a radio group, prior to a user clicking buttons ,  you may  need to set the checked value of whichever button is checked by default. This code fragment from radio.r shows how to do it. 
	
	Button1->checked = TRUE
	display RadioGroup1, MyTextBoxGroup

The three buttons in our example each have the property "checked value " and "unchecked value" . They also have a field attached , rbvalue1 , rbvalue2 and rbvalue3. When the program user clicks on a button the associated field is assigned to the corrected checked value. As this is a radiogroup, checking one button automatically unchecks the others so their uncheck value comes into play and is assigned to their associated fields. 

[bookmark: _Toc82954830]Editing properties at run time using "->" 

Button1 , is one of the three buttons making up the group . It has a property ,checked, which may be set by your program using the reference operator (minus greater than sign )  "->" .	
More generally the reference operator may be use to reference any property of any control. For example, 

		Textbox1->style = WS_INVISIBLE 
		wintask->style     = WS_CLOSE 
		Textbox1->data_font = "Times New Roman@14"
The properties being set here are all the same as the ones that could be set by using spd but doing things this way allows you to change properties while the program is running. You can find out the names of properties to use in your program by looking up "property clauses" in the reference manual.  

We have an event function defined for each button. They are all the same function in this case. If you want to respond to a radio button  press , use the event_enable property  EV_BUTTON_DOWN.  
If you look at the event function in the example, evButtonDown(), you will see we use the system function nameof() and display it as a caption. As you would guess, this simply returns a value giving the name of the object passed as parameter.  

[bookmark: _Toc82954831]Checkboxes, checkbox groups
Look at samples/checkbox.r and checkbox.h in conjunction with the following text 

Check boxes behave in the same way as radio groups except that checking one button in the group does not uncheck other members of the group.  Strictly, therefore, it is not necessary for checkboxes to be in groups at all. However, that is the way the system works in any case. 

In the example , we have a field defined for the group and for each button. The fields associated with the individual buttons work in exactly the same way as for radio buttons.   Notice the event_enable property for each button . It is defined as 
		event_enable = EV_BUTTON_DOWN | EV_BUTTON_UP
The button down event checks the button the button up event unchecks it. We need both events to be handled by our program because the unchecks are not happening automatically as they did with radio buttons.  

If you have a field defined for the group, as opposed to each button,  its value will be set according to scheme based on which buttons in the group are set.  See the reference manual for a full explanation. This will be recognisable  to many programmers who are familiar with binary bit masks. 

[bookmark: _Toc82954832]Listboxes

List boxes are list of values set out in a vertical line with a scroll bar if required. A value in a list box is selected by clicking on it. Sculptor list boxes come in one of 3 types which can be set in the listbox type property in spd .These are :-

LIST_EDIT    : folded up drop down list with option to add new member                     to list. Often known as a combo box.
LIST_BUTTON  : same as LIST_EDIT  by no optional edit
LIST_BOX     : unfolded list long enough to fit in specified length of listbox. 

[bookmark: _Toc82954833]Adding values to a list box 
We use the function :-
		list_box_add(lbName,ValueToAdd,LineToAddTo)

If the line number is left at 0 the value is just added to the end of the list. 

Eg to add 3 values to an empty listbox :-

		list_box_add(MyListBox,"LARGE",0)
		list_box_add(MyListBox,"MEDIUM",0)
		list_box_add(MyListBox,"SMALL",0)

If you need to clear a list box - use the clear command as you would with any control.

[bookmark: _Toc82954834]Finding out which values have been selected 

After a list box has been edited you can find out which line(s) have been selected by using the listbox_get_selected() function. 

Assume temps defined as 

!temp lbvalue,, a20
!temp lineno ,, i2
!temp npick  ,, i2

then 
	npick = listbox_get_selected(MyListBox, lbvalue, lineno)

npick is set to the number of values selected in the list box. lbvalue is set to the text selected, and lineno to the line number of the selected value. If the list box is of type LIST_BOX and the mode property of the listbox is set to allow multiple selection,  npick may be more than one. Look at the style=EM_MULTIPLE property for the first list box in the example . 

 In the case of EM_MULTIPLE use arrays for lbvalue and lineno which are big enough to receive all possible values selected. 

In  sample/listbox.r  we have the same  event function evShowListBoxes() for each type of list box and show the value selected for each list box in the textboxes to the right of the listboxes . In the case of the second list box , we just set the value of the number selected not each of the values. 
[bookmark: _Toc82954835]Setting/Unsetting  a selected value 
If , for example , in initialisation code, we want to set a value as selected/unselected we use set_selected() 

		set_selected(lbMyListBox,2,TRUE) 


This displays line 2 as selected. Use TRUE to select, FALSE to unselect. 

List boxes have a number of functions to support them - see the on-line manual for a full list – just search on listbox  in the index. 

 
[bookmark: _Toc82954836]Exercises 6

Exercise 1 –  button groups 
Copy chap05/ex4.r (.h, .i)  to chap06/ex1.r (.h, .i) and add in the following improvements :-

i) Find out your username ( sys.UserName )  and use this information to make an entry in the passwd file and set yourself up as an administrator.

ii) Bypass the password dialog if the current user is you, as defined in the last bit,  but still read in the record from the passwd file . ( Remember the read key =   construct ) .  Check that you still can get to the Edit Passwords option using your AdminUser status. 

Your username may have upper and lower case letters . Out passwd users are all upper case. The standard  function toupper(mystring)  which returns mystring as an uppercase string should be used to lookup passwd.UserName.  
 
iii) Change the Y/N answer for the admin user in the edit passwords routine to  use a checkbox instead of a textbox

iv) The customers are grouped in four regions :- UK, Europe, Africa, Asia and middle east . This is held in the field customer.Region . Use a radio group to manage this field. 



[bookmark: _Toc31882862][bookmark: _Toc33524187][bookmark: _Toc33525863][bookmark: _Toc35939785][bookmark: _Toc35963059][bookmark: _Toc36201850][bookmark: _Toc80111987][bookmark: _Toc80176037][bookmark: _Toc80631956][bookmark: _Toc80693606][bookmark: _Toc81407861][bookmark: _Toc81407925][bookmark: _Toc81909265][bookmark: _Toc81909338][bookmark: _Toc82093845][bookmark: _Toc82093924][bookmark: _Toc82332302][bookmark: _Toc82332384][bookmark: _Toc82621620][bookmark: _Toc82621716][bookmark: _Toc82947678][bookmark: _Toc82954757][bookmark: _Toc82954837] 
[bookmark: _Toc82954838]Chapter 7 Debugging. 

It is said that there are 2 types of program - trivial programs and programs with bugs in them. Sooner or later will need a debugger, so before moving on to some non-trivial programs we will run the Sculptor debugger. If you have used a debugger before you will know roughly what to expect. Generally check out the tool bar options at the top left corner. For a fuller treatment of the debugger see the reference manual under debugging.

Try using chap02/special.r as a test. Don't worry if you don't understand what the program is doing. All we want to do is watch it execute line by line, jump over and out of functions and set break points etc 

First of all compile your program in debug mode, with a -d
Eg
		scc -d special.r
Now run the program in debug mode
		scdebug special.q 

Anyone who has ever used a debugger will have no difficulty finding out how to set break points, examine variables , run to the next line or to a break point. 

To see the value of a variable hover the mouse over any referecne to ti in the source code. 

The only features we will discuss here are stepping and setting breakpoints. To step to the next line, over a function, out of a function etc you will see the icons in the tool bar . 

When the program starts , keep clicking the "step to next line" icon . When you get to a function call step over it using the next icon button long. Then keep clicking next   until you are inside a function. Click the step out button to carry on after the function has returned. 
[bookmark: _Toc31882864][bookmark: _Toc33524189][bookmark: _Toc33525865][bookmark: _Toc35939787][bookmark: _Toc35963061][bookmark: _Toc36201852][bookmark: _Toc80111989][bookmark: _Toc80176039][bookmark: _Toc80631958][bookmark: _Toc80693608][bookmark: _Toc81407863][bookmark: _Toc81407927][bookmark: _Toc81909267][bookmark: _Toc81909340][bookmark: _Toc82093847][bookmark: _Toc82093926][bookmark: _Toc82332304][bookmark: _Toc82332386]If you set/unset a break point click to the right of the line number in the column marked B. Lines marked in this way in the debugger will write a text string "_bp" to your source code at the breakpoint . This does not affect your code on any other way . 
A few closing remarks . Don't try to debug two programs at the same time. The system reads one program at a time.
It is well worth running all your program through the debugger before releasing them on an unsuspecting world . Even if they appear to work they may not be doing it in the way intended. Remember the adage about trivial programs.

If a program is hanging when you run the debugger you can press "Step to next statement" and it will stop at the next statement after the point it has got to – you will need this in the exercises.

[bookmark: _Toc82954839]Exercises 7  
You will find a  program ex1.r   in chap07. It is not a long program but it still  needs serious debugging . You could probably work out what is wrong quite easily just by looking at the source code but use the debugger as an exercise. 
Note that we declare a temp with type n8.0/my in the exercise. This is a money field and we will consider money further later on. For now just accept that money is dealt with in pennies not pounds. The amounts displayed will be in pennies but we will find out more about managing this later. 

Exercise 1 

i) Button 1 is supposed to find the customer with the highest balance. It puts out an ironic message saying "wait a minute .. "  but  then hangs indefinitely. 
ii) Even when Button 1 does not hang and gives the right balance, it still does not say who the customer was correctly.
iii) Button 2 finds the first item in the stock file which does not have a price set greater than zero . It works correctly the first time but if you run it again it comes back with nothing
iv) Button 3 is supposed to look up the price of the item code given give but does nothing. 
v) Button 4 is supposed to add up the balances of all the customers and show the total but it gives a different total every time it is run. At best,  only one of the totals can be right. 



[bookmark: _Toc82954840]Chapter 8 - More on data types, operators , formats, string manipulation, system variables 

For the rest of this chapter we keep samples/dtypes/ to hand which contains programs showing the points raised here. 

[bookmark: _Toc82954841]The modulus operator 

Like most languages Sculptor has a modulus operator "%" . We will need to be familiar with this in what follows . If we had integers x,y and z then
		z = x % y  
means that z is assigned the remainder on division of x / y

Examples
		6 % 3 = 0       ( 2 remainder 0 ) 
		11 % 4 = 3     ( 2 remainder 3 ) 
		6 % 5 = 5       ( 0 remainder 5 ) 
		etc 
The calculator on windows has a mod operator which is the same thing. 
	
[bookmark: _Toc82954842]Integer division
If we have integers x,y and z then 
	z = x / y
will always return an integer . 
eg in integer division 
	9 / 4 = 2   ( remainder lost ) 
If x and y remain integers but z is a real number then expression 

	z = x / y 
will still assign a whole number to z because the left side of the expression is still an integer division. If now x or y was an r8 this would work as expected , ie the true value, not the truncated value,  would be assigned to z. You can force the left hand side to be a real number as in the follwing example. 

!temp x ,, i4  = 14
!temp y ,, i4  = 4
!temp z ,, r8 

	z = x / y              /* z = 3.0 */
	z = 1.0 * x / y     /* z = 3.5 */

	

  	
[bookmark: _Toc82954843]Dates 
In samples/dtypes there is a program dates.r which should be read in conjunction with this section. Run it with the debugger on . Step through the initialisation code and type in values for any fields on screen which allow it and follow the code into the validation routines.
If you use ddeditor on the stockitem file you will see that a date is defined as an i4 ( 4 byte integer) and then with its logical type set to Date . If you want to define a date in your program use the form :-

	!temp MyDate ,, i4/dn 

The "dn" indicates that this integer is to be treated as a day number. Sculptor stores dates in  4 byte integers which count the days from 1/1/0001. If you asssign sys.Date to a standard i4 you will see a value of approximately 365 * Current Year.   You can utilise the fact that dates are integers in your date calculations. Eg 
Dates may be treated as integers or as strings according to your needs . 

	MyDate = "1/2/2021"         /* assign a string */
	MyDate = sys.Date + 7      /* add like an integer */

A date can be assigned with a string thus :-

			
It is best, for the sake of clarity,  to avoid 2 digit years but if you do then sensible rules will apply. This assignment will do what you would expect.  
MyDate = "1/1/21"       /* assign 1/1/2021  */ 
There are rules for resolving two digit years entered into a text box or assigned as above. See the reference manual under conversion of 2-digit year input to 4 digits for a full explanation. 
	 
Dates conveniently work out so that on divsion by 7 the remainder gives the day of the week as a number . 

			Monday = 1		
Tues   = 2
			...
			Sat    = 6
			Sun    = 0 

Eg  

		if ( MyDate % 7 = 5 ) {
			prompt "Thank God it's Friday"
}



If you need to decompose a date into day, month, year components, or the reverse,  you can use 2 standard functions decdate() and encdate() 

Example – decdate() decompose date type into day, month, year
 
!temp MyDate,,i4/dn
!temp MyDay,,i1
!temp MyMonth,,i1
!temp MyYear,,i4

     MyDate  = "1/2/2023" 
     decdate (MyDay, MyMonth, MyYear, MyDate)

The temps are set to the appropriate values. 
	MyDay  = 1
	MyMonth = 2
	MyYear = 2023


Example :- encdate() to encode a Day,Month,Year field into a date day number :-

		MyDay    = 1
		MyMonth  = 2
		Year   = 2023

		encdate(MyDate,Day,Month,Year) 

MyDate now has the value "1/2/2023"  . Remember Sculptor stores the date as an integer so if you copied MyDate to MyInt , MyInt would show an integer value of  738552 ( days since 1/1/0001 )

In dates.r , if you type in one of the day,month,year integers you will see that a validation function for these variables is called.  It then uses encdate() on a local temp ,  ldate of type i4/dn. If this field gets set to zero by the encdate() then you know the day,month,year combination was invalid . eg 29/2/2021 or 31/9/2021 etc 
 

[bookmark: _Toc82954844]Date formats dd/mm/yy / dd/mm/yyyy etc etc etc 

Dates mostly follow simple intuitive rules when you set a format in a data dictionary or for a temp in a program . The format will be used by the system when it displays data ( but not in the prompt command as at version 6.3 ) . See the reference manual for a full list of date formats . Here are a few examples which you will see when you run  dates.r 

!temp Date1 ,, i4/dn,"dd/mm/yyyy"
!temp Date2 ,, i4/dn,"dd-mm-yy"
!temp Date3 ,, i4/dn,"DD/MM/YY"
!temp USDate ,, i4/dn, "mm/dd/yyyyy" 

If these dates were all set to 3/9/2021 they would display as :

	 3/ 9/2021
3- 9-21
03/09/21
09/03/2021


[bookmark: _Toc82954845]Date and time format 
The dates.r sample program has a temp defined thus :

!temp tdatetime, "Date and time" , i4/td , "HH:MM, dd/yy/yyyy"

You will notice the type qualifier if /td , not /dn meaning that this is a date/time field . Setting 
	tdatetime = sys.Date 
will set the HH:MM part of the date/time to zero. As you will see in the program code the correct system variable to use is 
	tdatetime  =  sys.SysTime 

Sculptor provides a multitude of time only formats. See the refence manual for a complete list. 

[bookmark: _Toc82954846]String manipulation 

Assume we have 3 strings :-

!temp s1 ,, a5
!temp s2 ,, a5
!temp s3 ,, a10

When you make an assignment to a string it is padded out with spaces, if required, by the length of the string being assign to 

Eg

		s1 = "ABC"

Actually sets s1 to "ABC  " ( ABC + 2 spaces ) . Make sure to use double quotes for string literals , single quotes get unexpected results. 

As we have noted previously, Sculptor is very free and easy about data types and will always try to something obvious and helpful if data types do not match.  If you assign a long string to a short string it will be shortened to fit. 

Eg

		s1 = "ABCDEFHIJKLM"

is the same as 

		s1 = "ABCDE"

You can add strings together to form new strings using the "+" and "/" operators. Using the plus operator operand 1 is added to operand 2 including the trailing spaces of operand 1.


Eg

	s1 = "ABC"
	s2 = "DEF" 
	s3 = s1 + s2 

If you display s3 you will see that it contains "ABC  DEF". If you want the trailing spaces cutting off then use the "/" operator

Eg

	s3 = s1 / s2 
	
yields s3 = "ABCDEF" . 

Other useful string functions are left(), right() , centre(), toupper(), tolower(), strlen() , tostr() ,  instr() and setstr() and more . The latter three are dealt with below. However ,  look them up in the reference manual  as well. There are other string handling functions - look them up under string functions in general. 
When forming formatted strings derived from other non-string data types use the tostr() function which converts a non-string data type to a string if a simple assignment does not work in the format you require. 

Eg

!temp MyInt ,, i4
!temp MyDate ,, i4/dn
!temp MyStr  ,, a100

	MyInt = 1234
	MyDate = sys.Date

Simply assigning 
	MyString = MyInt 
	MyString = MyDate 

will work in a default format. If you require more control however, use tostr() 

	Mystr = tostr(MyInt,"#####00") 
	MyStr = tostr(MyInt,"####")
	MyStr = tostr(MyDate,"dd.mm.yyyy")
	Mystr = "Date and int " + tostr(MyDate) + " " + tostr(MyInt )

etc

[bookmark: _Toc82954847]String functions  instr(),  setstr|() and  getstr()

Look at samples/dtypes/string.r in conjunction with the following notes. Use the debugger to trace what the program does.

instr(MyString1,MyInt,MyString2)

returns the value of the place in Mystring1 where MyString2 is to be found starting the search at MyInt. 
eg

	

	prompt instr("XXXHelloXXX", 1, "Hello") 
	prompt instr("XXXHelloXXX", 3, "Hello") 
	prompt instr("XXXHelloXXX", 5, "Hello")

In order, these prompts display the values 4,4 and zero. 

The getstr(MyString,pos,len) function,  returns a string which  starts in MyStr at character position pos for len characters. 


eg 
prompt getstr("XXXHelloXXX",4,5 ) 
	prompt getstr("XXXHelloXXX",1,3 )

will display "Hello" and then "XXX" 

The function  setstr(dest, pos, len , source) copies  characters from source into dest but only len characters starting at pos in dest 


!temp MyStr1 ,, a32 = "XXXHelloXXX"
!temp MyStr2 ,, a32 = "Bye. Back next Tuesday"

	setstr(MyStr1,4,5,MyStr2) 
	prompt MyStr1

this displays "XXXBye. XXX" , ie take 3 only the 1str 3 characters from string2 and copy them into string1 without disturbing the other characters. 

[bookmark: _Toc82954848]String comparisons with > , < , <>, = 
These work in a natural way as long as you take account of case sensitivity. Remember "A" – "Z" all come before "a" in the ASCII code table. You can compare single character or long strings – they all work telephone book style as long as you stock to a single case. 
In the exercises we will need to see if a character in a string is a digit so make a note of the example below. 
Examples :

	if  ( "A"  <=  "B" ) prompt "Yes" 
if  ( "4"  <=  "5" ) prompt "Yes"
	if  ("FRED" > "BILL" ) prompt "Yes" 
if  ("IAIN" > "IAN" ) prompt "Yes" 
	
This should get four yes prompts . 

	

 

[bookmark: _Toc82954849]Money 

Money data can be stored in 4 or 8 byte format. The only difference is the range of values which can be stored. Money fields are declared thus 


!temp MyMoney1 ,, n4.0/my
!temp MyMoney2 ,, n8.0/my 

In the ddeeditor , set the field type to "number" and select 4 or 8 bytes. 

When you assign a literal value to a money field be careful to do it in pennies ( or lower unit of currency ).

Eg if you assign 

		MyMoney = 12.34 

When you display the money field you will see 0.12. The correct assignment is :-

		MyMoney = 1234

If a money values is typed into a textbox it is, of course, typed in with a decimal point as a user would expect. 


[bookmark: _Toc82954850]System variables


There are many system variables which live in a record structure called "sys". We have already seen sys.Date and sys.Traps and other.  Some of the most commonly used are :-

		sys.Date
		sys.Time
		sys.Traps
		sys.Arg 
		sys.UserName 

sys.Arg is an array of alphas which contains the parameters to a call to srepwc. Thus if you call your program :-
		srepwc MyProg param1 param2 param3 

sys.Arg[] contains :-

		sys.Arg[1] = "srepwc"
		sys.Arg[2] = "MyProg"
		sys.Arg[3] = "param1"
		sys.Arg[4] = "param2"


[bookmark: _Toc82954851]Exercises 8


Exercise 1 
In working/chap08 you will find ex1.r which needs fixing. 

What is displayed when you run ex1 ? If you were expecting "ABCD",  how do you fix it to get this ?

Exercise 2 
In working/chap08 you will find ex2.r which needs fixing. It assigns an initial value of 123 to x and then tries to output a message to say :

	"x=123" 
or whatever the value of x is at the time. However, the program fails because we mix the types integer and string in the prompt statement. This can be fixed. 
 

Exercise 3 Write a valid post code function 

In the UK valid post codes have two tokens ( or words )  :

A postcode  may be one of the forms : 

AA9A 9AA 
A9A  9AA 
A9   9AA 
A99  9AA 
AA9  9AA 
AA99 9AA

Eg    SW1 2EN  is of the form AA9 9AA 
 
Use spd to create ex3.r as a starter program. Then ... 

i) Declare a record structure PostCode which has 3 alpha fields , postcode, tok1, tok2 . 
ii) Create a text box for  postcode and the two tokens ,a  default OK button , a validation routine ValCheckPostCode() attached to the textbox for postcode and an ExitProg() function . 

iii) Write a function Split() which splits the postcode into tok1, tok2 which are the two separate words(tokens)  of postcode. 

iv) Write a function ToFormat(tstr) which takes a token such as "WC1" and converts it to "AA9" – its general format

v) Write a function IsPostCode() which uses Split() and ToFormat() to  return  TRUE or FALSE if both tokens satisfy the rules of UK post codes.

Take a look at SimpleProgs/function2.r  !function CheckPostCodeFormat() if you get stuck at this point.

vi) When the user enters a post code and presses return the validation function should call IsPostCode() and advise the user as to whether the post code is valid or not. 




